Common Source Amplifier: NMOS Inverter Amplifier with PMOS Current Load.

Figure 1(a) is a common source amplifier with ideal current source load. Figure 1(b) is its implementation using PMOS with constant gate voltage.

1. Low Frequency Small Signal Equivalent Circuit

Figure 1(c) shows its low frequency equivalent circuit. That is, all the stray capacitances are ignored. The two-port parameters are to be determined. Comparing Figure 1(c) and Figure 1(d), one obtains:

\[ Y_L = g_{ds2} \quad \text{(or } Z_L = r_{ds2}) \quad ; \quad Y_S = \infty \quad \text{(or } Z_S = 0) \]

\[ V_1 = V_i = V_{gs1} \]

and

\[ V_2 = V_o \]

From Fig 1, the current equations are derived to obtain the Y parameters:

\[ I_1 = 0 \]

\[ I_2 = g_{m1} V_1 + g_{ds1} V_2 \]

That is, the Y-parameter matrix is given by:

\[
Y = \begin{bmatrix}
0 & 0 \\
g_{m} & g_{ds1}
\end{bmatrix} \quad ; \quad \text{det}Y = 0
\]

The input impedance of common source circuit,

\[ Z_i = \frac{g_{ds1} + g_{ds2}}{\text{det}Y + y_{11} Y_L} = \frac{g_{ds1} + g_{ds2}}{0 + (0)g_{ds2}} = \infty \]

The output impedance of the common source circuit,

\[ Z_o = \frac{1}{\text{det}Y + y_{22} Y_S} = \frac{1}{y_{22}} = \frac{1}{g_{ds1}} \]

The dc voltage gain is,

\[ A_{v0} = -\frac{y_{21}}{y_{22} + Y_L} = -\frac{g_{m1}}{g_{ds1} + g_{ds2}} \quad \text{or} \]

\[ A_{v0} = -g_{m1} \left( Z_o / Z_L \right) = -g_{m1} R_{\text{out}} \quad ; \quad R_{\text{out}} = Z_o / Z_L = \frac{1}{g_{ds1} + g_{ds2}} \]

The current gain is,
\[ A_1 = -\frac{y_{21} Y_l}{\det Y + y_{11} Y_l} = -\frac{g_{m1} g_{n2}}{0 + 0} = \infty \]
Figure 1. Common Source Amplifier: (a) ideal current source load, (b) PMOS current source load, (c) low frequency small signal equivalent circuit, and (d) two-port representation.
2. Common Source NMOS Inverter Amplifier with PMOS Current Load Static Static Characteristic

The small signal equivalent circuit assumes that its operating point has been properly set. To achieve this, one needs to determine the static or large signal characteristics of the amplifier. It will be shown that the operating point range for a high gain amplifier is very narrow. Hence, its selection is very critical. A little error will cause the circuit to cease functioning as amplifier.

Figure 2 shows the various regions of operation for each transistor. It is determined as follows:

**MN Transistor Operating Regions:**
- **Cutoff**
  \[ V_{GSN} = V_{IN} - V_{SS} < V_{TN} \]
  \[ V_{IN} < V_{TN} + V_{SS} = 1 + (-2.5) = -1.5V \]
- **Saturation**
  \[ V_{IN} > -1.5 \]
  and
  \[ V_{GSN} - V_{TN} < V_{DSN} \]
  \[ V_{IN} - V_{SS} - V_{TN} < V_{O} - V_{SS} \]
  \[ V_{IN} - V_{TN} < V_{O} \]
  \[ V_{IN} - 1 < V_{O} \]
- **Ohmic**
  \[ V_{IN} > -1.5 \]
  and
  \[ V_{IN} - V_{TN} > V_{O} \]

**MP Transistor (Current Source) Operating Regions:**
- **Cutoff**
  \[ |V_{GSP}| = |V_{G} - V_{DD}| < |V_{TP}| \]
  \[ |0 - 2.5| < |1| \] **Can’t be satisfied \( \Rightarrow \) No cutoff region
- **Saturation**
  \[ |V_{GSP}| - |V_{TP}| < |V_{DSP}| \]
  \[ |V_{G} - V_{DD}| - |V_{TP}| < |V_{O} - V_{DD}| \]
  \[ (V_{DD} - V_{G}) + V_{TP} < (V_{DD} - V_{O}) \]
  \[-V_{G} + V_{TP} < -V_{O} \]
  \[ V_{G} - V_{TP} > V_{O} \]
  \[ 0 - (-1) > V_{O} \]
  \[ V_{O} < 1 \]
- **Ohmic**
  \[ V_{O} > 1 \]
These are summarized as follows:

<table>
<thead>
<tr>
<th>Operating Region</th>
<th>MN</th>
<th>MP</th>
</tr>
</thead>
<tbody>
<tr>
<td>Cutoff</td>
<td>$V_{IN}&lt;-1.5$</td>
<td>None</td>
</tr>
<tr>
<td>Saturation</td>
<td>$V_{IN}&gt;-1.5&amp; V_{IN}^-&lt;V_O$</td>
<td>$V_O&lt;1$</td>
</tr>
<tr>
<td>Ohmic</td>
<td>$V_{IN}&gt;-1.5&amp; V_{IN}^-&gt;V_O$</td>
<td>$V_O&gt;1$</td>
</tr>
</tbody>
</table>

The various regions of operation of inverting amplifier are summarized as follows:

<table>
<thead>
<tr>
<th>Operating Region</th>
<th>MN</th>
<th>MP</th>
</tr>
</thead>
<tbody>
<tr>
<td>Region I</td>
<td>Cutoff</td>
<td>Ohmic</td>
</tr>
<tr>
<td>Region II</td>
<td>Saturation</td>
<td>Ohmic</td>
</tr>
<tr>
<td>Region III</td>
<td>Saturation</td>
<td>Saturation</td>
</tr>
<tr>
<td>Region IV</td>
<td>Ohmic</td>
<td>Saturation</td>
</tr>
<tr>
<td>Region V</td>
<td>Ohmic</td>
<td>No Cutoff, always on</td>
</tr>
</tbody>
</table>

These regions are shown in the in the PSpice transfer characteristic graph.

The common source amplifier circuit is shown in Figure 1(b). The power supplies and netlist used in PSpice simulation and numerical calculations are also indicated. From Figure 1(b), the load current is equal to the driver current, i.e.

$$i_P = i_N$$

The PMOS transistor MP with a constant $V_{GS}$ behaves as a constant current source. Subscripts P and N are used to distinguish the variables and parameters of the two types of transistors. The current equation of the transistor MP in saturated case is given by:

$$i_P = \beta P \left[ V_{PP} - V_{DS} \right]$$ when $V_{DS} - |V_{TP}| \leq 0$

Substituting for $V_{GS} = V_G - V_{DD}$,

$$i_P = \beta P \left[ (V_{DD} - V_G) + V_{TP} \right]$$ when $(V_{DD} - V_G) + V_{TP} \leq V_{DD} - V_O$ or $V_G - V_{TP} > V_O$

Similarly, the PMOS current equation for the ohmic case is given by:

$$i_P = \beta P \left[ (V_{GG} - |V_{TP}|) |V_{DS} - |V_{DP}| / 2 \right]$$ when $V_G - V_{TP} \leq V_O$

For the NMOS driver transistor MN current equations are given by:
\[ i_n = \frac{(\beta_n)}{2}(V_{GSN} - V_{TN})^2 \text{ when } V_{GSN} - V_{TN} \leq V_{DSN}, \text{ MN is saturated.} \]

Substituting for \( V_{GSN} = V_i - V_{SS} \), and \( V_{DSN} = V_O - V_{SS} \)
\[ i_n = \frac{(\beta_N)}{2}(V_i - V_{ss} - V_{TN})^2 \text{ when } V_i - V_{ss} - V_{TN} \leq V_o - V_{ss} \text{ or } V_i - V_{TN} \leq V_o, \text{ MN is saturated.} \]

Similarly, the NMOS current equation for the ohmic case is given by:
\[ i_n = (V_i - V_{ss} - V_{TN})(V_o - V_{ss}) - (V_o - V_{SS})^2 / 2 \text{ when } V_i - V_{TN} > V_o, \text{ MN is ohmic.} \]

The static characteristic is determined by equating the corresponding current equations at each voltage range:

1. \( V_i - V_{ss} < V_{TN} \) and \( V_G - V_{TP} = 0 \). The driver transistor MN is off, and MP is on (ohmic). Hence \( V_o = V_{dd} \).

2. \( V_i - V_{SS} > V_{TN} \), \( V_i - V_{TN} < V_o \), and \( V_G - V_{TP} < V_o \). The driver transistor MN is saturated, and the load transistor MP is ohmic.
\[ (\beta_n / 2)(V_i - V_{ss} - V_{TN})^2 = \beta_p \left((V_{dd} - V_G + V_{TP})(V_{dd} - V_o) - (V_{dd} - V_o)^2 / 2 \right] \]

Solving for \( V_o \),
\[ V_o = (V_G - V_{TP}) - \sqrt{(V_{dd} - V_G + V_{TP})^2 - \beta_p (V_i - V_{ss} - V_{TN})^2} \]
where \( \beta_R = \beta_N / \beta_P \)

3. \( V_i - V_{SS} > V_{TN} \), \( V_i - V_{TN} < V_o \), and \( V_G - V_{TP} > V_o \). The driver transistor MN is saturated, and the load transistor MP is also saturated.
\[ (\beta_n / 2)(V_i - V_{ss} - V_{TN})^2 = \beta_p / 2\left((V_{dd} - V_G + V_{TP})^2 \right] \]

This equation is independent of \( V_o \), it is used to determine the operating point \( V_{bias} \) of the inverter.

Solving for \( V_{bias} \),
\[ V_i = V_{bias} = V_{ss} + V_{TN} + (V_{dd} - V_G + V_{TP}) / \sqrt{\beta_R} \]

This value is only an approximation. The exact value is obtained from the PSpice simulation. The bias voltage corresponds to the input voltage \( V_i \) when the output voltage \( V_o = (V_{dd} + V_{SS}) / 2 = (2.5 - 2.5) / 2 = 0 \).

4. \( V_i - V_{SS} > V_{TN} \), \( V_i - V_{TN} > V_O \), and \( V_G - V_{TP} > V_o \). The driver transistor MN is ohmic, and the load transistor MP is saturated.
\[ \beta_N \left((V_i - V_{ss} - V_{TN})(V_o - V_{ss}) - (V_o - V_{SS})^2 / 2 \right] = \beta_p / 2 \left((V_{dd} - V_G + V_{TP})^2 \right] \]

Solving for \( V_o \),
\[ V_o = (V_i - V_{TN}) - \sqrt{(V_i - V_{ss} - V_{TN})^2 - (V_{dd} - V_G + V_{TP})^2 / \beta_R} \]

5. \( V_o = V_{SS} \), when the load transistor MP is off. It occurs when \( V_{GSP} = V_G - V_{dd} > V_{TP} \). But for the given \( V_G (=0) \) this can not happen. This is verified in the Pspice simulation. Figure 2 shows that \( V_{SS} = -2.5 \) V can never be reached.
Figure 2. The various operating regions of nmos common source amplifier with pmos current load

For the given parameters, the operating point $V_{bias}$ is computed for the given example. The W/L ratio must use the $L_{eff} = L - 2*LD = 5.4u - 2*(0.5u) = 4.4u$. 

*PSpice file for NMOS Inverter with PMOS Current Load
*Filename="Lab3.cir"
VIN 1 0 DC 0VOLT AC 1V
VDD 3 0 DC 2.5VOLT
VSS 4 0 DC -2.5VOLT
VG 5 0 DC 0VOLT
M1 2 1 4 4 MN W=9.6U L=5.4U
M2 2 5 3 3 MP W=25.8U L=5.4U

.MODEL MN NMOS VTO=1 KP=40U
+ GAMMA=1.0 LAMBDA=0.02 PHI=0.6
+ TOX=0.05U LD=0.5U CJ=5E-4 CJSW=10E-10
+ U0=550 MJ=0.5 MJSW=0.5 CGSO=0.4E-9 CGDO=0.4E-9
.MODEL MP PMOS VTO=-1 KP=15U
+ GAMMA=0.6 LAMBDA=0.02 PHI=0.6
+ TOX=0.05U LD=0.5U CJ=5E-4 CJSW=10E-10
+ U0=200 MJ=0.5 MJSW=0.5 CGSO=0.4E-9 CGDO=0.4E-9
*Analysis
.DC VIN -2.5 2.5 0.05
.TF V(2) VIN
.PROBE
.END
\[ \beta_R = \frac{K_N (W_N / L_{Neff})}{K_p (W_{Peff} / L_{Peff})} = \frac{(40E - 6)(9.6E - 6/4.4E - 6)}{(15E - 6)(25.8E - 6/4.4E - 6)} = 1 \]

\[ V_i = V_{bias} = V_{SS} + V_{TN} + (V_{DD} - V_G + V_{TP})/\sqrt{\beta_R} = -2.5 + 1 + (2.5 - 0 - 1)/\sqrt{1} = 0 \]

The bias point from PSpice simulation is 7.6205mV reasonably closed to the calculated value of 0. From the transfer characteristic graph, a bias voltage of 0 is a reasonably good choice, it will be used in the subsequent experiments.

Figure 1(c) shows the low frequency small signal equivalent circuit. The small signal dc gain is given in section 1, as

\[ A_{v0} = -g_{m1} R_{out} \]

where:

\[ R_{out} = Z_O/Z_L = \frac{1}{g_{ds1} + g_{ds2}} \]

In the analysis above, we used the N and P notation to distinguish the two-type of transistor used in the common source amplifier. In Figure 1(b), MN is M1 and MP is M2

The small signal dc voltage gain is translated to:

\[ A_v = \frac{v_o}{v_i} = -g_{mN} (R_{ON} // R_{OP}) \]

where:

\[ R_{ON} = 1/(\lambda_N I_{DSQ}) \]
\[ R_{OP} = 1/(\lambda_P I_{DSQ}) \]
\[ g_{mN} = \sqrt{2\beta_N I_{DSQ}} \]
\[ I_{DSQ} = I_N = I_P = (\beta_N / 2)(V_{bias} - V_{SS} - V_{TN})^2 \]

\[ \beta_N = K_N (W_N / L_{N}) = (40E - 6)(9.6u/4.4u) = 87.3 \text{ uA/V}^2 \]
\[ I_{DSQ} = (\beta_N / 2)(V_{bias} - V_{SS} - V_{TN})^2 = (87.3E - 6/2)(0 - (-2.5) - 1)^2 = 98.21 \text{ uA} \]
\[ g_{mN} = \sqrt{2\beta_N I_{DSQ}} = \sqrt{2(87.3E - 6)(98.21E - 6)} = 130.95 \text{ umho} \]
\[ R_{ON} = 1/(\lambda_N I_{DSQ}) = 1/[(.02)(98.21E - 6)] = 0.509M \]
\[ R_{OP} = 1/(\lambda_P I_{DSQ}) = 1/[(.02)(98.21E - 6)] = 0.509M \]
\[ A_v = -g_{mN} (R_{ON} // R_{OP}) = -(130.95E - 6)(.509M//.509M) = -33.18 \]
The low frequency input resistance $R_{in} = \infty$, since the input is capacitive. The output resistance $R_{out} = (R_{ON}//R_{OP}) = .2545M$. These calculations agree well with PSpice simulation results of:

**** SMALL-SIGNAL CHARACTERISTICS

$V(2)/VIN = -3.500E+01$

INPUT RESISTANCE AT VIN = 1.000E+20

OUTPUT RESISTANCE AT V(2) = 2.536E+05

3. High Frequency Small Signal Equivalent Circuit

![High Frequency Small Signal Equivalent Circuit](image)

Figure 3. Common source amplifier parasitic capacitances.

Figure 3 shows all the parasitic capacitances in the common source amplifier. Figure 4 shows the high frequency small signal equivalent circuit of the common source amplifier circuit. Comparing Figure 4(b) and 4(c) one obtains:
The current equation is:

\[ I_1 = sC_{gs1}V_1 + sC_{gd1}(V_1 - V_2) = s(C_{gs1} + C_{gdi})V_1 - sC_{gdi}V_2 \]
\[ I_2 = g_{ml}V_1 + sC_{gdi}(V_2 - V_1) + (g_{ds1} + sC_o)V_2 = (g_{ml} - sC_{gdi})V_1 + [g_{ds1} + s(C_{gdi} + C_o)]V_2 \]

Figure 4. The high frequency equivalent circuit: (a) All the parasitic capacitances, (b) Combining capacitances, and (c) two-port representation.

The corresponding Y-parameter matrix is:
The overall voltage gain is:

\[
Y = \begin{bmatrix}
    s(C_{gs1} + C_{gd1}) & -sC_{gd1} \\
    g_{m1} - sC_{gd1} & g_{ds1} + s(C_{gd1} + C_o)
\end{bmatrix}
\]

\[
\det Y = s(C_{gs1} + C_{gd1})[g_{ds1} + s(C_{gd1} + C_o)] + sC_{gd1}(g_{m1} - sC_{gd1})
\]

\[
= s(g_{ds1}C_{gs1} + g_{ds1}C_{gd1} + g_{m1}C_{gd1}) + s^2(C_{gs1}C_{gd1} + C_{gs1}C_o + C_{gd1}C_o)
\]

The overall voltage gain is:

\[
A_V = \frac{V_2}{V_1} = -\frac{y_{21}}{y_{22} + Y_L} = -\frac{g_{m1} - sC_{gd1}}{g_{ds1} + s(C_{gd1} + C_o) + g_{ds2}} = -\frac{g_{m1} - sC_{gd1}}{(g_{ds1} + g_{ds2}) + s(C_{gd1} + C_o)}
\]

where:

\[
A_{V0} = -g_{m1}R_{out}; R_{out} = \frac{1}{g_{ds1} + g_{ds2}}
\]

\[
z_1 = \frac{g_{m1}}{C_{gd1}}
\]

\[
p_1 = w_{BW} = -\frac{1}{R_{out}C_{out}}; C_{out} = C_{gd1} + C_o = C_{gd1} + C_{gd2} + C_{db1} + C_{db2} + C_L
\]

\[
w_{GBW} = A_{V0}w_{BW} = (-g_{m1}R_{out})\left(-\frac{1}{R_{out}C_{out}}\right) = \frac{g_{m1}}{C_{out}}
\]

That is, the \( w_{GBW} \) is directly proportional to the transconductance of the driver transistor and inversely proportional to the total output capacitance. The Bode Plot is shown in Figure 5.
Experiments on Common Source Amplifier Circuits

1. Pspice Simulation Using the Netlist of NMOS Inverter with PMOS Current Load Including Cgs and Cgd Transistor Parasitic Capacitances Only.

Figure 1(b) is simulated with PSpice. Its netlist file is shown below. No area and perimeter are supplied for the source and drain. That is, PSpice will only compute the $C_{GS}$ and $C_{GD}$ in the simulation. This will be verified using the small signal circuit including $C_{GS}$ and $C_{GD}$ only, discuss in section 3.

*PSpice file for NMOS Inverter with PMOS Current Load
*Filename="Lab3.cir"
VIN 1 0 DC 0VOLT AC 1V
VDD 3 0 DC 2.5VOLT
VSS 4 0 DC -2.5VOLT

Figure 5. Approximate bode plot of common source amplifier.
VG 5 0 DC 0VOLT
M1 2 1 4 4 MN W=9.6U L=5.4U
M2 2 5 3 3 MP W=25.8U L=5.4U

.MODEL MN NMOS VTO=1 KP=40U
+ GAMMA=1.0 LAMBDA=0.02 PHI=0.6
+ TOX=0.05U LD=0.5U CJ=5E-4 CJSW=10E-10
+ U0=550 MJ=0.5 MJSW=0.5 CGSO=0.4E-9 CGDO=0.4E-9
.MODEL MP PMOS VTO=-1 KP=15U
+ GAMMA=0.6 LAMBDA=0.02 PHI=0.6
+ TOX=0.05U LD=0.5U CJ=5E-4 CJSW=10E-10
+ U0=200 MJ=0.5 MJSW=0.5 CGSO=0.4E-9 CGDO=0.4E-9

*Analysis
.DC VIN -2.5 2.5 0.05
.TF V(2) VIN
.AC DEC 100 1HZ 10GHZ
.PROBE
.END

The extracted parameters are:

\[ A_v = A_{v0} - 3 \quad f_{BW} = 44.218\text{M} \]
\[ A_v = 0 \text{ db} \quad f_{GBW} = 1.6127\text{G} \]
\[ PM = 75.826^\circ \]

The Bode Plot is shown below:
2. Pspice Simulation Using the Netlist of NMOS Inverter with PMOS Current Load Including All Transistor Parasitic Capacitances.

This is achieved by supplying the area and perimeter of the source and drain in the Pspice netlist.

*PSpice file for NMOS Inverter with PMOS Current Load
*Filename="Lab31.cir"
VIN 1 0 DC 0VOLT AC 1V
VDD 3 0 DC 2.5VOLT
VSS 4 0 DC -2.5VOLT
VG 5 0 DC 0VOLT
M1 2 1 4 4 MN W=9.6U L=5.4U AD=40.32P AS=40.32P PD=27.6U PS=27.6U
M2 2 5 3 3 MP W=25.8U L=5.4U AD=108.36P AS=108.36P PD=60U PS=60U

.MODEL MN NMOS VTO=1 KP=40U
+ GAMMA=1.0 LAMBDA=0.02 PHI=0.6
+ TOX=0.05U LD=0.5U CJ=5E-4 CJSW=10E-10
+ U0=550 MJ=0.5 MJSW=0.5 CGSO=0.4E-9 CGDO=0.4E-9
.MODEL MP PMOS VTO=-1 KP=15U
+ GAMMA=0.6 LAMBDA=0.02 PHI=0.6
+ TOX=0.05U LD=0.5U CJ=5E-4 CJSW=10E-10
+ U0=200 MJ=0.5 MJSW=0.5 CGSO=0.4E-9 CGDO=0.4E-9
*Analysis
.DC VIN -2.5 2.5 0.05
The extracted parameters are:

\[ A_V = A_{v0} - 3 \quad f_{BW} = 6.6672M \]
\[ A_V = 0 \text{ db} \quad f_{GBW} = 223.441M \]

The Bode Plot is shown below:


The following MATLAB m file has been written to calculate the parasitic capacitance for each transistor by specifying W, L, VBD, and VBS.

function \([\text{CGS}, \text{CGD}, \text{CBD}, \text{CBS}] = \text{cap(W,L,VBD,VBS)}\)  
% Mfile must be save under /MATLAB directory  
% SPICE PARAMETERS  
% TOX = 0.05; \quad \% 0.05U  
% COX = 0.69; \quad \% 0.69 fF/\mu m^2 COX=(3.9e0/TOX)  
% CGSO = 0.4; \quad \% 0.4E-9 F/m=0.4fF/\mu m  
% CGDO = 0.4; \quad \% 0.4E-9 F/m=0.4fF/\mu m  
% CI = 0.5; \quad \% 5E-4 F/m^2=0.5fF/\mu m^2  
% CJSW = 1; \quad \% 10E-10 F/m = 1fF/\mu m
LD = 0.5  % 0.5U
MJ = 0.5;
MJSW = 0.5;
PHI=0.6;  % 0.5 V

%SPICE PARAMETERS END
LEFF=L-2*LD;
LMIN = 4.2;  % 7*(.6)=4.2;4.5u actual measurement from LAB3 layout
CGC = COX * W * LEFF;
CGSOX = CGSO * W;
CGDOX = CGDO * W;
CGS = CGSOX + (2/3) * CGC;
CGD = CGDOX;
AD = W * LMIN;
PD = 2*(W + LMIN);
AS = W * LMIN;
PS = 2*(W + LMIN);
CBDJ = CJ/(1 - VBD/PHI)^MJ;
CBDJSW = CJSW/(1 - VBD/PHI)^MJSW;
CBD = AD*CBDJ + PD*CBDJSW;
CBDJSW = CJSW/(1 - VBD/PHI)^MJSW;

The following is a sample call using the circuit for lab3.

For the NMOS transistor,

\[ [CGS, CGD, CBD, CBS] = \text{cap}(9.6, 5.4, -2.5, 0) \]
\[ CGS = 23.2704 \quad CGD = 3.8400 \quad CBD = 21.0116 \quad CBS = 61.8400 \]

For the PMOS transistor,

\[ [CGS, CGD, CBD, CBS] = \text{cap}(25.8, 5.4, -2.5, 0) \]
\[ CGS = 62.5392 \quad CGD = 10.3200 \quad CBD = 50.2325 \quad CBS = 152.0200 \]

\[ C_0 = C_{DBn} + C_{DBp} + C_{DGp} = 21.0116 + 50.2325 + 10.32 = 81.5641 \text{ fF} \]
\[ CGSn = 23.2704 \text{ fF} \]
\[ CGDn = 3.8400 \text{ fF} \]
* Small signal equivalent circuit of NMOS inverter with PMOS
* Current load. Only CGS CGD are calculated
* Filename=lab3eqn.cir"
Vin 1 0 AC 1V
Rin 1 0 1E+20
Cgd 1 2 3.84fF
Cgs 1 0 23.27fF
Gm 2 0 1 0 130.95u
Ro 2 0 .2545Meg
Co 2 0 10.32ff
.TF V(2) Vin
.AC DEC 100 1HZ 10GHZ
.PROBE
.END

The extracted parameters are:

\[ A_V = A_{V_0} - 3 \quad f_{BW} = 42.341M \]
\[ A_V = 0 \text{ db} \quad f_{GBW} = 1.5493G \]

Comparing with theoretical calculation:

\[ w_{GBW} = \frac{g_m}{C_o} = \frac{130.95E - 6}{10.32E - 15} = 12.69 \times 10^9 \]
\[ f_{GBW} = \frac{w_{GBW}}{2\pi} = \frac{12.69E9}{2\pi} = 2.02 \times 10^9 = 2.02G \]
\[ w_{BW} = \frac{1}{R_o C_o} = \frac{1}{(.2545E6)(10.32E - 15)} = 0.381 \times 10^9 \]
\[ f_{BW} = \frac{w_{BW}}{2\pi} = \frac{0.38E9}{2\pi} = 60 \times 10^6 = 60M \]

The Bode Plot is shown below:
4. Pspice Simulation Using Small Signal Equivalent Circuit Including All Transistor Parasitic Capacitances.

* Small signal equivalent circuit of NMOS inverter with PMOS
* Current load.
* Filename = "lab3eq.cir"
Vin 1 0 AC 1V
Rin 1 0 1E+20
Cgd 1 2 3.84fF
Cgs 1 0 23.27fF
Gm 2 0 1 0 130.95u
Ro 2 0 .2545Meg
Co 2 0 81.5641fF
.TF V(2) Vin
.AC DEC 100 1HZ 10GHZ
.PROBE
.END

The extracted parameters are:

\[ A_v = A_v0 - 3 \quad f_{BW} = 7.3485M \]
\[ A_v = 0 \text{ db} \quad f_{GBW} = 235.957M \]

Comparing with theoretical calculation:
Design Specification

Design a common source amplifier with a gain of $A_v=50$ and output impedance of $R_o=1$ Meg.

$$R_o = R_{ON}/R_{OP}$$

Assume $R_{ON}=R_{OP}$, hence

$$R_{ON} = R_{OP} = 2R_o = 2\text{Meg}$$
\[ R_{\text{ON}} = \frac{1}{\lambda I_{\text{SQ}}} = R_{\text{OP}} = \frac{1}{\lambda I_{\text{SQ}}} \]

Solving for \( I_{\text{DSQ}} \),

\[ I_{\text{DSQ}} = \frac{1}{\lambda R_{\text{ON}}} = \frac{1}{\lambda R_{\text{OP}}} = \frac{1}{(0.02)(2\times10^6)} = 25 \mu A \]

From the gain equation the \((W/L)\), can be obtained as follows:

\[ A_v = -g_{mn} R_{O} = -\sqrt{2\beta N I_{\text{DSQ}}} R_{O} = -\sqrt{2K_N (W_N/L_{Neff}) I_{\text{DSQ}}} R_{O} \]

\[ (W_N/L_{Neff}) = \frac{(A_v/R_{O})^2}{2K_N I_{\text{DSQ}}} = \frac{(50/1\times10^6)^2}{2(40\times10^{-6})(25\times10^{-6})} = \frac{5}{4} \]

Using technology with \( \lambda = 0.6 \mu \), the \( W, L \) are selected as follows:

\[ W_N = 5\lambda = 5(0.6) = 3 \mu \]

\[ L_{Neff} = 4\lambda = 4(0.6) = 2.4 \mu \]

\[ L_N = L_{Neff} + 2L_D = 2.4 \mu + 2(0.5 \mu) = 3.4 \mu \]

The drawn length must be multiple of \( \lambda \), hence \( L_N = 3.6 \mu (= 6 \lambda) \), the nearest computed value must be selected.

The \( W, L \) of the PMOS are computed to achieve \( \beta_R = 1 \), that is

\[ \beta_R = \frac{K_p (W_p/L_{Peff})}{K_N (W_N/L_{Neff})} = \frac{K_p W_p}{K_N W_N} = 1 \text{ ; since } L_{Neff} = L_{Peff} \]

\[ W_p = \frac{K_N W_N}{K_p} = \frac{(40\times10^{-6})(3\mu)}{(15\times10^{-6})} = 8 \mu \text{; select } 7.8 \mu (= 13 \lambda) \]

\[ L_p = L_N = 3.6 \mu \]
Figure 1. Common source amplifier using current mirror to establish the bias current.

The common source amplifier is initially simulated using Pspice with Vbias set to 0 to obtain the DC transfer characteristic. The actual Vbias is then located at the center of the transition region, region 3 of DC transfer characteristic.

*PSpice file for NMOS Inverter with PMOS Current Load
*Filename="Lab3tst.cir"
VIN 1 0 DC 0VOLT AC 1V
VDD 3 0 DC 2.5VOLT
VSS 4 0 DC -2.5VOLT
M1 2 4 1 MN W=3U L=3.6U
M2 2 3 5 MP W=7.8U L=3.6U
M3 5 3 3 MP W=7.8U L=3.6U
IB 5 4 25U

.MODEL MN NMOS VTO=1 KP=40U
+ GAMMA=1.0 LAMBDA=0.02 PHI=0.6
+ TOX=0.05U LD=0.5U CJ=5E-4 CJSW=10E-10
+ U0=550 MJ=0.5 MJSW=0.5 CGSO=0.4E-9 CGDO=0.4E-9
.MODEL MP PMOS VTO=-1 KP=15U
+ GAMMA=0.6 LAMBDA=0.02 PHI=0.6
+ TOX=0.05U LD=0.5U CJ=5E-4 CJSW=10E-10
+ U0=200 MJ=0.5 MJSW=0.5 CGSO=0.4E-9 CGDO=0.4E-9
*Analysis
The simulation results are for Vbias=0. Hence the following small signal characteristics are not what we are looking for.

**** SMALL-SIGNAL CHARACTERISTICS

\[ \frac{V(2)}{VIN} = -4.066\times10^{-1} \]

INPUT RESISTANCE AT VIN = 1.000E+20

OUTPUT RESISTANCE AT V(2) = 1.986E+04

The above simulation results will give us the node voltage \( V_5 = V_G \).

<table>
<thead>
<tr>
<th>NODE</th>
<th>VOLTAGE</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>0.0000</td>
</tr>
<tr>
<td>2</td>
<td>-2.0602</td>
</tr>
<tr>
<td>3</td>
<td>2.5000</td>
</tr>
<tr>
<td>4</td>
<td>-2.5000</td>
</tr>
<tr>
<td>5</td>
<td>0.4667</td>
</tr>
</tbody>
</table>

With node voltage \( V_G \) known, the approximate Vbias can be calculated as follows:

\[
V_{bias} = V_{SS} + V_{TN} + (V_{DD} - V_G + V_{TP})/\sqrt{\beta R} = -2.5 + 1 + (2.5 - V_G - 1)/\sqrt{\beta} = -V_G
\]

Vbias = \( V_5 = V_G = -0.4667 \)

The node voltage \( V_G \) can also be calculated theoretically, this will be shown in the next section.

To obtain the desired small signal characteristics, the Vbias must be set correctly by changing one line in the Pspice netlist.

\[ VIN 1 0 DC -0.4667VOLT AC 1V \]

The small signal characteristics show that the design specifications are achieved within 5%: \( Av = -50.8 \), \( R_o = 1.027 \text{Meg} \).
V(2)/VIN = -5.080E+01

INPUT RESISTANCE AT VIN = 1.000E+20

OUTPUT RESISTANCE AT V(2) = 1.027E+06

NODE VOLTAGE | NODE VOLTAGE | NODE VOLTAGE | NODE VOLTAGE
-------------|--------------|--------------|--------------
(1) -0.4667 | (2) -0.6650  | (3) 2.5000   | (4) -2.5000  |
(5) 0.4667

**Increasing The Gain of Common Source Amplifier**

The new specification is to increased the gain to \( \text{Av}=70 \), and maintaining \( \text{Ro}=1\text{Meg} \).

Using the current mirror principle the \( I_{DSQ}=25\mu\text{A}=I_{bias} \) can be maintained by passing \( I_{bias} \) to M3. With a constant \( I_{DSQ} \), \( \text{Ro} \) will also be constant. The new gain can be achieved as follows:

\[
A_{V2} = \frac{-g_{m2}R_O}{g_{m1}R_O} = \frac{g_{m2}}{g_{m1}} = \frac{\sqrt{2K_N(W_{N2}/L_{Neff})I_{DSQ}}}{\sqrt{2K_N(W_{N1}/L_{Neff})I_{DSQ}}} = \frac{\sqrt{W_{N2}}}{\sqrt{W_{N1}}}
\]

\[
W_{N2} = \left(\frac{A_{V2}}{A_{V1}}\right)^2 W_{N1} = \left(\frac{70}{50}\right)^2 3u = \frac{49}{25} 3u \approx 6u
\]

**Theoretical Calculation of \( V_G \)**

PMOS transistor is in saturation whenever,

\[
|V_{GSP}| - |V_{TP}| < |V_{DSP}|
\]  

--- (1)

For PMOS transistor M3, the drain is connected to gate, hence eq (1) is always satisfied. M3 is always operating in saturation mode. Hence, the drain current is given by:

\[
I_{bias} = I_{P3} = (\beta_p/2)(V_G - V_{DD})^2 = (\beta_p/2)(1.5 - V_G)^2
\]  

Solving for \( V_G \),
\[ V_G = 1.5 - \frac{2I_{bias}}{\beta_R} = 1.5 - \frac{2I_{bias}}{K_p(W_p/L_{peff})} = 1.5 - \frac{2(25E-6)}{(15E-6)(7.8/2.6)} = 0.4459 \]

Simulation result is 0.4667

Changing the W/L of M1 to increase the gain will alter the \( \beta_R \) ratio.

\[ \beta_R = \frac{K_N(W_N/L_{Neff})}{K_p(W_p/L_{peff})} = \frac{K_N W_N}{K_p W_p} = \frac{(40E-6)(6E-6)}{(15E-6)(7.8E-6)} = 2.05 \]

The new theoretical \( V_{bias} \) is calculated,

\[ V_{bias} = V_{ss} + V_{TN} + (V_{DD} - V_G + V_{TP})/\sqrt{\beta_R} = -2.5 + 1 + (2.5 - 0.4459 - 1)/\sqrt{2.05} = -0.764 \]

Again the actual \( V_{bias} \) is determined from the DC transfer characteristics, \( V_{bias} = -0.777 \) which is very close to the theoretical calculation above.

The new circuit is simulated by changing the following two lines in the Pspice netlist:

VIN 1 0 DC –0.777VOLT AC 1V

M1 2 1 4 4 MN W=6U L=3.6U

The simulation results show that the desired small signal characteristics are within 5%, \( A_V = -72.61 \), and \( R_o = 1.038 \text{Meg} \)

**** SMALL-SIGNAL CHARACTERISTICS

\[ V(2)/VIN = -7.261E+01 \]

INPUT RESISTANCE AT VIN = 1.000E+20

OUTPUT RESISTANCE AT V(2) = 1.038E+06

 NODE VOLTAGE NODE VOLTAGE NODE VOLTAGE NODE VOLTAGE
( 1)  -0.7770  (2)  -0.1122  ( 3)  2.5000  ( 4)  -2.5000
( 5)  0.4667
Replace the current source by resistor R

Figure 2. Replacing IB by R

In Figure 2,

\[ I_{bias} = \frac{V_G - V_{SS}}{R} = \frac{V_G + 2.5}{R} \]

\[ R = \frac{V_G + 0.4667 + 2.5}{25E - 6} = 0.119\text{Meg} \approx 120k \]

To simulate the new circuit, the Pspice netlist is modified as follows:

IB 5 4 25U (Delete this line)

R 5 4 120k (Add this line)

Simulation results are very close with the circuit with current source.

**** SMALL-SIGNAL CHARACTERISTICS
V(2)/VIN = -7.261E+01

INPUT RESISTANCE AT VIN = 1.000E+20

OUTPUT RESISTANCE AT V(2) = 1.043E+06

NODE VOLTAGE NODE VOLTAGE NODE VOLTAGE NODE VOLTAGE NODE VOLTAGE

( 1)  - .7770 ( 2)  - .3597 ( 3)  2.5000 ( 4)  -2.5000
( 5)  .4716

The Effect of Eliminating Parasitic Capacitance Parameters

Eliminate terms in the MODEL that refers to parasitic capacitances. The new Pspice netlist is shown below. The simulation results show that the bandwidth of amplifier becomes infinite.

* PSpice file for NMOS Inverter with PMOS Current Load
* Filename="Lab3tst1.cir"
VIN 1 0 DC -0.777VOLT AC 1V
VDD 3 0 DC 2.5VOLT
VSS 4 0 DC -2.5VOLT

M1 2 1 4 4 MN W=6U L=3.6U
M2 2 5 3 3 MP W=7.8U L=3.6U
M3 5 5 3 3 MP W=7.8U L=3.6U
R 5 4 120k
*IB 5 4 25U

.MODEL MN NMOS VTO=1 KP=40U
+ GAMMA=1.0 LAMBDA=0.02 PHI=0.6
+ TOX=0.05U LD=0.5U
.MODEL MP PMOS VTO=-1 KP=15U
+ GAMMA=0.6 LAMBDA=0.02 PHI=0.6
+ TOX=0.05U LD=0.5U

*Analysis
.DC VIN -2.5 2.5 0.05
.TF V(2) VIN
.AC DEC 100 1HZ 1000GHZ
.PROBE
.END